
KATEDRA INFORMATIKY
UNIVERZITA PALACKÉHO V OLOMOUCI
www.inf.upol.cz

Michal Krupka
krupka.inf.upol.cz
michal.krupka@upol.cz
17. listopadu 12, 771 46 Olomouc

Paradigmata programování 3 ◇ poznámky k přednášce

1. Objekty a třídy

verze z 23. září 2020

1 Úvod

V tomto semestru se budeme zabývat důležitým programovacím paradigmatem, objektově orientovaným programováním (OOP). Stejně jako v minulých semestrech bude platit to, co jsem říkal před rokem na první přednášce:

Cílem čtyřsemestrálního kurzu Paradigmata programování je seznámit studenty s principy programování a základními přístupy a styly v programování používanými (tzv. paradigmaty*). Budeme to dělat bez ohledu na to, jaké přístupy se v současné době zrovna používají v praxi — zkušenost říká, že ty se rychle mění a vyvíjejí, zatímco principy zůstávají mnoho desetiletí stále stejné.*

Jedině programátor, který rozumí principům, na kterých je programování a programovací jazyky postaveno, je schopen programovat kvalitně.

Dávat důraz na principy mimo jiné znamená

- *Nepostupovat metodou pokus/omyl*
- *Vědět vždy přesně, co se v počítači děje...*
- *... až do té míry, že jsme schopni předpovídat výsledky.*

Přístup k objektově orientovanému programování se v různých jazycích více nebo méně liší. Jelikož v tomto předmětu klademe důraz na *principy*, nebudeme se méně důležitými prvky jednotlivých jazyků zabývat. Jako dříve nám i teď půjde o hlubší pochopení principů které jsou obecné a do všech (nebo většiny) objektově orientovaných jazyků se promítají. Podrobné informace o konkrétním programovacím jazyce rozšířeném v praxi se dozvíte v jiném souběžném předmětu.

2 Objekty jako datové struktury

V minulých semestrech jsme se s datovými strukturami už setkali. *Datovou strukturu* jsme chápali jako hodnotu, která obsahuje jednu nebo více jiných hodnot. K práci s datovými strukturami se používají tři základní typy funkcí:

1. **Konstruktory.** To jsou funkce, které novou datovou strukturu vytváří. Obvykle jako argumenty akceptují hodnoty, které se do nové struktury uloží.

2. **Selektory.** Funkce, které pro datovou strukturu vrací jednu z hodnot ve struktuře uložených.

3. **Mutátory.** Umožňují měnit hodnoty uložené ve struktuře. Ve funkcionálním programovacím stylu se nepoužívají.

Poznámky:

- To, že datové struktury jsou hodnoty, znamená, že je lze ukládat do proměnných, používat jako argumenty funkcí a že je funkce mohou vracet jako svůj výsledek.
- Přesnější název pro datové struktury tak, jak jsme je používali, je *abstraktní datová struktura*. Struktura je abstraktní, když uživatel nezná její implementaci a k práci s ní používá jen selektory a mutátory (a novou strukturu vytváří konstruktorem).
- Každý typ datové struktury má obvykle svou vlastní sadu konstruktorů, selektorů a mutátorů.

Princip objektově orientovaného přístupu v programování spočívá v tom, že selektory, mutátory a další funkce pracující s datovou strukturou chápeme jako její součást. Objektově orientovaný pohled tedy chápe datové struktury jako samostatné a nezávislé jednotky, které samy (na daný pokyn) vykonávají potřebné operace se svým obsahem.

To lze považovat za základní charakteristiku objektově orientovaného přístupu: data a kód se chápou jako jeden celek.

Datové struktury chápané společně s jejich funkcemi objektově orientované programování nazývá *objekty*. Programovací jazyky, které umožňují (a někdy až vynucují) objektově orientovaný pohled na programování, obsahují různé nástroje, které práci s takto chápanými objekty usnadňují.

S objektově orientovaným programováním je neoddělitelně spjatý nový typ problému, se kterým jsme se dosud nesetkali a který je ve větších programech vždy potřeba řešit: problém organizace zdrojového kódu, který se v OOP promítá do problému jak rozdělit činnosti vykonávané programem mezi objekty, neboli, jak rozhodnout, který objekt je zodpovědný za kterou činnost.

3 Objekty

V objektově orientovaném programování je běžící program složen ze samostatných a nezávislých jednotek, zvaných *objekty*.[1] Každý objekt je zodpovědný za určitou

[1] Prvním objektově orientovaným jazykem byl jazyk Simula 67, vytvořený v 60. letech minulého století v Norském výpočetním středisku v Oslu. Jak i název napovídá, jazyk byl navržen tak, aby byl vhodný pro simulace reálných procesů. Z reality je převzatá základní myšlenka programu jako souhrnu vzájemně komunikujících objektů. Velmi důležitým objektovým programovacím jazykem

část činnosti programu. Koná ji na základě pokynů, které mu udělujeme zasíláním *zpráv*. V reakci na přijetí zprávy objekt vykoná požadovanou činnost. Ta se obvykle týká jeho vlastního obsahu.

Zopakujme znovu dvě základní charakteristiky objektů: *samostatnost* a *nezávislost*.

Samostatnost a nezávislost objektů

Samostatnost objektů znamená, že jim stačí říct, *co* mají dělat, není nutné specifikovat *jak*. Objekt by měl být pro programátora, který ho používá, co nejjednodušší k použití, měl by od něj vyžadovat co nejmenší znalosti a odstínit ho od nepodstatných technických detailů řešení problému.

Nezávislost Objekt je tím nezávislejší, čím méně ke své práci vyžaduje existenci vnějších faktorů. Nezávislost objektů umožňuje jejich *znovupoužitelnost*, neboli zachování funkčnosti po přenesení do jiného prostředí (programu).

Tyto dvě charakteristiky si dobře zapamatujte. V budoucnu je můžeme používat k posuzování kvality napsaného programu.

Zprávy zasílané objektům jsou identifikovány jménem. Součástí zprávy mohou být i argumenty, podobně jako u volání funkce. Výsledkem zaslání zprávy může být (podobně jako u volání funkce) vrácená hodnota.

Na přijetí zprávy objekt reaguje tak, že spustí kód, který vykoná činnost, jež je po něm zprávou požadována. Tomuto kódu se říká *metoda*. Metoda je zvláštní druh funkce, má jméno a parametry. Každý objekt může obsahovat více metod, podle toho, jaké zprávy může přijímat. Po přijetí zprávy se spustí metoda stejného jména, jaké má přijímaná zpráva, a spustí se s těmi argumenty, se kterými byla zpráva poslána. Metody mají stejně jako funkce návratovou hodnotu (hodnoty). Ta je nakonec výsledkem zaslání zprávy.

Metodě spuštěné v reakci na přijetí zprávy se říká *obsluha zprávy*. Vykonání obsluhy zprávy se říká její *obsloužení*.

Metody lze chápat jako kód, který se nachází *uvnitř objektu* (tuhle věc ještě za chvíli upřesníme).

Objekty mají *vnitřní stav*. To je souhrn dat, která objekt obsahuje jakožto datová struktura. V reakci na přijetí zprávy může objekt svůj vnitřní stav změnit. To je kromě vrácení hodnoty další efekt, který může zaslání zprávy objektu mít.

Data v objektech jsou (podobně jako u struktur v jazyce C) rozdělena do pojmenovaných položek, kterým budeme říkat *sloty*. Z pohledu uživatele objektu je

je SmallTalk, který vznikl začátkem 70. let v Xerox Palo Alto Research Center. Objektový systém Common Lispu (Common Lisp Object System, CLOS), kterému se budeme věnovat v tomto textu, vznikl ze starších objektových systémů CommonLOOPS a MIT Flavors. Common Lisp je historicky prvním standardizovaným programovacím jazykem používajícím objektově orientovaný přístup (ANSI standard).

ale jedno, jak je vnitřní stav v objektu reprezentován, protože uživatel s objektem komunikuje pouze zasíláním zpráv. Nezajímá ho, jak je napsaná metoda, která se v reakci na přijetí zprávy spustí, ani jak a s jakými daty uvnitř objektu pracuje.

Při psaní programu musíme důsledně odlišovat mezi autorem metod a vnitřní stavby objektu a *uživatelem objektu*, tj. programátorem, který objekt ve svém programu používá, a to bez ohledu na to, že to může být tentýž člověk. Uživatel objektu nemusí znát informace o implementaci objektu a neměl by být nucen dělat činnosti, které by měl objekt zvládnout sám. Tento důležitý princip rozebereme podrobně později.

4 Třídy

Podobně jako jiná data, i objekty mohou být různých typů. V objektově orientovaném programování se pro základní typy objektů používá pojem *třídy*. Na tomto místě uvedeme zjednodušenou definici třídy, kterou v dalších částech rozšíříme.

Aby dva objekty patřily téže třídě, musí splňovat tyto podmínky:

1. musí obsahovat stejnou sadu slotů, tedy stejný počet slotů stejných názvů (hodnoty těchto slotů však mohou být různé),
2. musí obsahovat stejné metody.

Definice třídy ve zdrojovém textu programu tyto dva údaje (kromě názvu třídy) uvádí. Třídu lze tedy chápat, jako popis objektu: obsahuje jednak seznam názvů jeho slotů a jednak definici všech jeho metod. Při běhu programu pak třída slouží jako předloha k vytváření nových objektů. A konečně, jak už jsem napsal nahoře, třída se také chápe jako datový typ, tedy množina objektů.

Objekt, který patří třídě, se nazývá její *instancí*. Chceme-li v programu vytvořit objekt, musíme mít pro něj definovánu třídu a v programu objekt vytvořit jako její instanci.

5 Třídy a instance v Common Lispu

V Common Lispu, přesněji v jeho objektovém systému CLOS, se nové třídy definují pomocí makra `defclass`, které specifikuje seznam slotů třídy, a pomocí makra `defmethod`, které slouží k definici metod instancí třídy.

Nové objekty se vytvářejí pomocí funkce `make-instance`. Ke čtení hodnoty slotu objektu slouží funkce s názvem `slot-value`, kterou lze v kombinaci s operátorem `setf` použít i k nastavování hodnot slotů.

Zprávy se v Common Lispu objektům zasílají pomocí stejné syntaxe, jakou se v tomto jazyce volají funkce.

Makro `defclass`

Zjednodušená syntax makra `defclass` je následující:

```
(defclass name () slots)

  name: symbol (nevyhodnocuje se)
  slots: seznam symbolů (nevyhodnocuje se)
```

Symbol *name* je název nově definované třídy, symboly ze seznamu *slots* jsou názvy slotů této třídy.

Prázdný seznam za symbolem *name* je součástí zjednodušené syntaxe. V dalších kapitolách, až se dozvíme více o třídách, ukážeme, co lze použít místo něj.

Příklad: třída `point`

Definice třídy `point`, jejíž instance by obsahovaly dva sloty s názvy `x` a `y`, by vypadala takto:

```
(defclass point ()
  (x y))
```

Funkce `make-instance`

Definovali-li jsme novou třídu (zatím bez metod, k jejichž definici se dostaneme vzápětí), měli bychom se naučit vytvářet její instance. V Common Lispu k tomu používáme funkci `make-instance`, jejíž zjednodušená syntax je tato:

```
(make-instance class-name)

  class-name: symbol
```

Funkce `make-instance` vytvoří a vrátí novou instanci třídy, jejíž jméno najde ve svém prvním parametru. Všechny sloty nově vytvořeného objektu jsou neinicializované a každý pokus získat jejich hodnotu skončí chybou (později si řekneme, jak se získávají hodnoty slotů a pak to budeme moci vyzkoušet).

Příklad: vytváření a prohlížení instancí

Vytvoření nové instance třídy `point` z předchozího příkladu:

```
CL-USER 2 > (make-instance 'point)
#<POINT 200DC0D7>
```

Pokud není jasné, proč jsme ve výrazu `(make-instance 'point)` symbol `point` kvotovali, je třeba si uvědomit, že `make-instance` je funkce a zopakovat si základy vyhodnocovacího procesu v Common Lispu.

Výsledek volání není příliš čitelný, v prostředí LispWorks si jej ale můžeme prohlédnout v inspektoru (Obr. 1).



Obrázek 1: Neinicializovaná instance třídy `point` v inspektoru

Text `#<unbound slot>` u názvů jednotlivých slotů znamená, že sloty jsou neinicializované. Můžeme jim ale pomocí prostředí LispWorks zkusit nastavit hodnotu. Klikneme-li na některý ze zobrazených slotů pravým tlačítkem, můžeme si v objevivší se nabídce vybrat volbu "Slots->Set..." tak, jak je znázorněno na Obrázku 2 a novou hodnotu slotu nastavit.

Funkce `slot-value`

K programovému čtení hodnot slotů slouží funkce `slot-value`, k jejich nastavování symbol `slot-value` v kombinaci s makrem `setf`. Syntax je následující:

`(slot-value` *`object slot-name`*`)`

`object:` objekt
`slot-name:` symbol

Příklad: práce s funkcí `slot-value`

Vytvořme instanci třídy `point` a uložme ji do proměnné `pt`:

```
CL-USER 2 > (setf pt (make-instance 'point))
#<POINT 216C0213>
```



Obrázek 2: Nastavování hodnoty slotu v inspektoru

Nyní zkusme získat hodnotu slotu `x` nově vytvořené instance:

```
CL-USER 3 > (slot-value pt 'x)

Error: The slot X is unbound in the object #<POINT 216C0213>
(an instance of class #<STANDARD-CLASS POINT 200972AB>).
  1 (continue) Try reading slot X again.
  2 Specify a value to use this time for slot X.
  3 Specify a value to set slot X to.
  4 (abort) Return to level 0.
  5 Return to top loop level 0.

Type :b for backtrace, :c <option number> to proceed, or :?
for other options
```

Vidíme, že došlo k chybě; slot `x` není v nově vytvořeném objektu inicializován. Z chybového stavu se dostaneme napsáním `:a` a zkusíme hodnotu slotu nejprve nastavit:

```
CL-USER 4 : 1 > :a

CL-USER 5 > (setf (slot-value pt 'x) 10)
10
```

Nyní již funkce `slot-value` chybu nevyvolá a vrátí nastavenou hodnotu:

```
CL-USER 6 > (slot-value pt 'x)
10
```

Nové hodnoty slotů lze také ověřit pomocí inspektoru (Obr 3).

K práci se sloty ještě dodejme, že používáním funkce `slot-value` ke čtení a nastavování hodnoty slotů vlastně porušujeme výše uvedený princip, že pro komunikaci s objekty se používají výhradně zprávy. K tomuto problému se dostaneme v další kapitole.

Makro `defmethod` a posílání zpráv

Zbývá vysvětlit, jak se v Common Lispu definují metody objektů a jak se objektům posílají zprávy. Jak jsme již řekli, všechny instance jedné třídy mají stejnou sadu metod. Metody jsou zvláštní druh funkce, proto se definují podobně. V Common Lispu je k definici metod připraveno makro `defmethod`. Jeho syntaxe (ve zjednodušené podobě, jak ji uvádíme na tomto místě), je stejná jako u makra `defun` s tou výjimkou, že u prvního parametru je třeba specifikovat jeho třídu. Tím se metoda definuje pro všechny instance třídy.



Obrázek 3: Instance třídy `point` po změně hodnoty slotu `x`

```
(defmethod message ((object class) arg1 arg2 ...)
  expr1
  expr2
  ... )

  message: symbol
  object: symbol
  class: symbol
  argi: symbol
  expri: výraz
```

Symbol *class* určuje třídu, pro jejíž instance metodu definujeme, symbol *message* současně název nové metody i název zprávy, kterou tato metoda obsluhuje. Výrazy *expr1*, *expr2* atd. tvoří *tělo metody*, které stejně jako tělo funkce definuje kód, který se provádí, když je metoda spuštěna. Symbol *object* je v těle metody navázán na objekt, jemuž byla zpráva poslána, symboly *arg1*, *arg2*, atd. na další argumenty, se kterými byla zpráva zaslána.

Jak již bylo řečeno, syntax zasílání zprávy je stejná jako syntax volání funkce:

```
(message object arg1 arg2 ...)

  message: symbol
  object: výraz
  argi: výraz
```

Symbol *message* musí být názvem zprávy, kterou lze zaslat objektu vzniklému vyhodnocením výrazu *object*. Zpráva je objektu zaslána s argumenty, vzniklými vyhodnocením výrazů *arg1*, *arg2* atd. Stejně jako u volání funkce jsou výrazy *object*, *arg1*, *arg2* atd. vyhodnoceny postupně zleva doprava.

Zasílání zprávy v Common Lispu má stejnou syntax jako volání funkce a ve skutečnosti opravdu o volání funkce jde. Jedná se o funkce zvláštního druhu, kterým se říká *generické funkce*. Tento fakt je dobré mít na paměti, protože vývojové prostředí o generických funkcích někdy (třeba při hlášení chyb) mluví. Ke konci semestru se ke zvláštnostem objektového systému Common Lispu dostaneme.

V mnoha jiných programovacích jazycích by se výše uvedené zaslání zprávy zapsalo takto:

```
object.message(arg1 arg2 ...)
```

Příklad: polární souřadnice

Řekněme, že potřebujeme zjišťovat polární souřadnice bodů. Správný objektový způsob řešení této úlohy je definovat nové zprávy, které budeme bodům k získání těchto informací zasílat.

Definujme tedy nové metody pro třídu `point`: metodu `r`, která bude vracet vzdálenost bodu od počátku (první složku jeho polárních souřadnic), a metodu `phi`, která bude vracet odchylku spojnice bodu a počátku od osy x (tedy druhou složku polárních souřadnic bodu)

Metoda `r` počítá vzdálenost bodu od počátku pomocí Pythagorovy věty:

```
(defmethod r ((point point))
  (let ((x (slot-value point 'x))
        (y (slot-value point 'y)))
    (sqrt (+ (* x x) (* y y)))))
```

Pokud nechápete, co znamená `(point point)` v této definici, podívejte se znovu na syntax makra `defmethod`. Zjistíte, že první položkou tohoto seznamu je symbol, na nějž bude při vykonávání těla metody navázán příjemce zprávy, zatímco druhou položkou je název třídy, pro jejíž instance metodu definujeme. Že jsou oba tyto symboly stejné, nevadí, v Common Lispu mohou být názvy proměnných a tříd stejné.

Po zaslání zprávy `r` bodu bychom tedy měli obdržet jeho vzdálenost od počátku. Vytvořme si na zkoušku instanci třídy `point` a nastavme jí hodnoty slotů `x` a `y` na `3` a `4`:

```
CL-USER 8 > (setf pt (make-instance 'point))
#<POINT 200BC6A3>

CL-USER 9 > (setf (slot-value pt 'x) 3
                  (slot-value pt 'y) 4)
4
```

Vytvořený objekt reprezentuje geometrický bod, který je znázorněn na Obrázku 4.



Obrázek 4: Bod o souřadnicích $(3, 4)$

Nyní zkusme získat vzdálenost tohoto bodu od počátku zasláním zprávy `r` naší instanci (připomeňme, že zprávy se objektům zasílají stejnou syntaxí jakou se volají funkce, tedy výraz `(r pt)` znamená zaslání zprávy `r` objektu `pt`):

```
CL-USER 10 > (r pt)
5.0
```

To správně, jelikož $\sqrt{3^2 + 4^2} = 5$.

Podobně definujme metodu `phi` (pochopení vyžaduje trochu matematických znalostí):

```
(defmethod phi ((point point))
  (let ((x (slot-value point 'x))
        (y (slot-value point 'y)))
    (cond ((> x 0) (atan (/ y x)))
          ((< x 0) (+ pi (atan (/ y x))))
          (t (* (signum y) (/ pi 2))))))
```

Další zkouška:

```
CL-USER 11 > (phi pt)
0.9272952
```

Tangens tohoto úhlu by měl být roven 4/3 (viz Obrázek 4):

```
CL-USER 12 > (tan *)
1.3333333
```

Příklad: nastavování polárních souřadnic

Definujme ještě metody pro nastavení polárních souřadnic bodu. Na rozdíl od předchozích budou tyto metody vyžadovat zadání argumentů. Vzhledem k tomu, že každá z nich mění obě kartézské souřadnice bodu současně, bude užitečné napsat nejprve metodu pro současné nastavení obou polárních souřadnic.

```
(defmethod set-r-phi ((point point) r phi)
  (setf (slot-value point 'x) (* r (cos phi))
        (slot-value point 'y) (* r (sin phi)))
  point)
```

Metody `set-r` a `set-phi` tuto metodu využijí (přesněji řečeno, zprávu `set-r-phi` zasílají):

```
(defmethod set-r ((point point) value)
  (set-r-phi point value (phi point)))

(defmethod set-phi ((point point) value)
  (set-r-phi point (r point) value))
```

Metody `set-r-phi`, `set-r` a `set-phi` vracejí vždy jako výsledek parametr `point`. Tento přístup budeme volit ve všech metodách, které mění stav objektu: vždy budeme jako výsledek vracet měněný objekt. Důvodem je, aby šlo objektu měnit více hodnot v jednom výrazu:

```
(set-r (set-phi pt pi) 1)
```

Nyní můžeme instancím třídy `point` posílat zprávy `set-r-phi`, `set-r` a `set-phi` a měnit tak jejich polární souřadnice. Vyzkoušejme to tak, že našemu bodu `pt` pošleme zprávu `set-phi` s argumentem `0`. Tím bychom měli zachovat jeho vzdálenost od počátku, ale odchylka od osy x by měla být nulová.

Zaslání zprávy `set-phi` s argumentem `0`:

```
CL-USER 13 > (set-phi pt 0)
#<POINT 200BC6A3>
```

Test polohy transformovaného bodu:

```
CL-USER 14 > (slot-value pt 'x)
5.0

CL-USER 15 > (slot-value pt 'y)
0.0
```

Výsledek je tedy podle očekávání (nová poloha bodu je na druhém konci modrého oblouku na Obrázku 4).

6 Inicializace slotů nových instancí

Ukažme si ještě jednu možnost makra `defclass`. V předchozích odstavcích jsme si všimli, že když vytvoříme novou instanci třídy, jsou všechny její sloty neinicializované a při pokusu o získání jejich hodnoty před jejím nastavením dojde k chybě. To se někdy nemusí hodit. Proto makro `defclass` stanovuje možnost, jak specifikovat počáteční hodnotu slotů nově vytvářené instance.

V obecnější podobě makra `defclass` je jeho syntax následující:

```
(defclass name () slots)

  name: symbol (nevyhodnocuje se)
  slots: seznam (nevyhodnocuje se)
```

Prvky seznamu *slots* mohou být symboly nebo seznamy. Je-li prvkem symbol, je jeho význam takový, jak již bylo řečeno, tedy specifikuje název slotu instancí třídy, který není při vzniku nové instance inicializován. Je-li prvkem tohoto seznamu seznam, musí být jeho tvar následující:

```
(slot-name :initform expr)

  slot-name: symbol
  expr: výraz
```

V tomto případě specifikuje symbol *slot-name* název definovaného slotu. Výraz *expr* je vyhodnocen pokaždé při vytváření nové instance třídy a jeho hodnota je do příslušného slotu instance uložena.

Sloty nově vytvořených instancí je dobré vždy hned inicializovat. Kromě uvedeného způsobu se později dozvíme další, vhodný na složitější situace.

Konzistence nově vytvořených instancí

Nově vytvořené instance by měly být rovnou v konzistentním stavu, tj. měly by splňovat všechny předpoklady kladené na instance dané třídy.

Toto pravidlo je užitečné z pohledu uživatele. K vytvoření nové funkční instance třídy mu stačí zavolat funkci `make-instance`. Nemusí provádět žádné další inicializace. Pro autora třídy znamená dodržení pravidla víc práce, ale tak to má být: programátor se má vždy snažit usnadňovat uživateli práci i za tu cenu, že jemu to práci přidělá.

Příklad: třída `point` s inicializací slotů

Základní předpoklad o bodech je bezesporu ten, že mají nějaké kartézské souřadnice. Proto upravíme definici třídy `point` tak, aby byly sloty `x` a `y` nových instancí inicializovány na hodnotu `0`:

```
(defclass point ()
  ((x :initform 0)
   (y :initform 0)))
```

Jak můžeme snadno zkusit, sloty nových instancí jsou nyní inicializovány:

```
CL-USER 1 > (setf pt (make-instance 'point))
#<POINT 20095117>

CL-USER 2 > (list (slot-value pt 'x) (slot-value pt 'y))
(0 0)
```

Příklad: třída `circle`

Nyní definujeme další třídu, jejíž instance budou reprezentovat geometrické útvary. Bude to třída `circle`. Jak známo, geometrie kruhu je určena jeho středem a poloměrem. Proto budou mít instance této třídy dva sloty. Slot `center`, který bude obsahovat instanci třídy `point` a slot `radius`, který bude obsahovat číslo. Každý z těchto slotů bude při vytvoření nové instance automaticky inicializován. Naším předpokladem kladeným na všechny kruhy totiž je, že jejich středem má být bod a poloměrem číslo.

```
(defclass circle ()
  ((center :initform (make-instance 'point))
   (radius :initform 1)))
```

Teď již necháme na čtenáři, aby si sám zkusil vytvořit novou instanci této třídy a prohlédl její sloty.

V následujícím příkladu ukážeme ještě několik chybových hlášení, se kterými se můžeme ve vývojovém prostředí LispWorks setkat.

Příklad: chybová hlášení

Pokud pošleme zprávu objektu, který pro ni nemá definovánu metodu (obsluhu této zprávy), dojde k chybě. Můžeme si to ukázat tak, že pošleme zprávu `phi` instanci třídy `circle`:

```
CL-USER 3 > (phi (make-instance 'circle))

Error: No applicable methods for #<STANDARD-GENERIC-FUNCTION PHI
21694CFA> with args (#<CIRCLE 216C3CF3>)
  1 (continue) Call #<STANDARD-GENERIC-FUNCTION PHI 21694CFA> again
  2 (abort) Return to level 0.
  3 Return to top loop level 0.

Type :b for backtrace, :c <option number> to proceed, or :? for
other options
```

V tomto hlášení o chybě je třeba všimnout si hlavně textu „No applicable methods“, který znamená, že jsme posílali zprávu objektu, který pro ni nemá definovanou obsluhu (metodu).

Vzhledem k tomu, že syntax zasílání zpráv je v Common Lispu stejná jako syntax volání funkce či aplikace jiného operátoru, nemohou se zprávy jmenovat stejně jako funkce, makra nebo speciální operátory. Proto následující definice vyvolá chybu (`set` je funkce Common Lispu):

```
CL-USER 5 > (defmethod set ((point point) coord value)
              (cond ((eql coord 'x) (set-x point value))
                    ((eql coord 'y) (set-y point value))))

Error: SET is defined as an ordinary function #<Function SET
202D54A2>
  1 (continue) Discard existing definition and create generic
function
  2 (abort) Return to level 0.
  3 Return to top loop level 0.

Type :b for backtrace, :c <option number> to proceed, or :? for
other options
```

Pokud bychom se pokusili poslat objektu zprávu, pro niž jsme nedefinovali metodu pro žádnou třídu, Common Lisp vůbec nepochopí, že se snažíme poslat zprávu, a bude volání interpretovat jako použití neexistujícího operátoru:

```
CL-USER 7 > (fí (make-instance 'circle))

Error: Undefined operator FÍ in form (FÍ (MAKE-INSTANCE (QUOTE
CIRCLE))).
  1 (continue) Try invoking FÍ again.
  2 Return some values from the form (FÍ (MAKE-INSTANCE (QUOTE
```

```
CIRCLE))).
  3 Try invoking something other than FÍ with the same arguments.
  4 Set the symbol-function of FÍ to another function.
  5 Set the macro-function of FÍ to another function.
  6 (abort) Return to level 0.
  7 Return to top loop level 0.

Type :b for backtrace or :c <option number> to proceed.
Type :bug-form "<subject>" for a bug report template or :? for
other options.
```

Otázky a úkoly na cvičení

Úkoly budou ještě doplněny.

1. Vysvětlete základní pojmy objektově orientovaného programování: objekt, třída, instance, zpráva, metoda, slot.

2. Elipsu v rovině lze zadat pomocí dvou ohnisek a jedné z poloos. Druhou poloosu už lze vždy dopočítat. Definujte třídu `ellipse`, která bude obsahovat sloty `focal-point-1` a `focal-point-2`. Sloty slouží k uchovávání dvou bodů, které jsou ohnisky elipsy. Volitelně můžete napsat metody pro přístup k těmto slotům (v níže uvedeném výpisu takové metody používáme; od příští kapitoly budete psát přístupové metody ke slotům povinně). Dále můžete ve třídě definovat další slot (sloty) na uchovávání potřebných informací o elipse.

3. Definujte pro třídu `ellipse` metody `major-semiaxis` a `minor-semiaxis`, které vrátí délku hlavní (to je vždy ta delší) a vedlejší poloosy

4. Definujte pro třídu `ellipse` metodu `current-center`, která vrátí novou instanci třídy `point` reprezentující střed elipsy.

5. Napište metodu `to-ellipse` třídy `circle`, která vrátí elipsu stejného tvaru, jako kruh, jenž je příjemcem zprávy.